

Lektion 4

Zeichnen von Kreisen und regelmäßigen Vielecken

Im Unterschied zur vorherigen Lektion geht es in dieser Lektion nicht darum, neue Programmierkonzepte und Rechnerbefehle zu lernen. Hier wollen wir etwas mehr über die Vielecke als geometrische Bilder erfahren und dadurch entdecken, wie wir sie mit Hilfe der Schildkröte zeichnen können.

Hinweis für die Lehrperson Diese Lektion dient eher der Festigung des bisherigen Stoffes und zur Entfaltung der Kreativität durch Erzeugung farbiger Bilder nach eigener Vorstellung. Für diese Lektion reichen üblicherweise 1 bis 2 Unterrichtsstunden.

Ein regelmäßiges k -Eck hat k Ecken und k gleich lange Seiten. Wenn du ein Vieleck, zum Beispiel ein 10-Eck, mit Bleistift zeichnen möchtest, musst du zehn Linien zeichnen und nach jeder Linie „ein bisschen“ die Richtung ändern (drehen).

Wie viel muss man drehen?

Kannst du so etwas berechnen?

Wenn man ein regelmäßiges Vieleck zeichnet, dreht man mehrmals aber am Ende steht man genau an der gleichen Stelle und schaut in genau die gleiche Richtung wie am Anfang (Abb. 4.1 auf der nächsten Seite)

Das bedeutet, dass man sich unterwegs volle 360° gedreht hat. Wenn man also ein regelmäßiges 10-Eck zeichnet, hat man sich genau zehn mal gedreht und zwar immer um




Abbildung 4.1

einen gleichgroßen Winkel. Also

$$\frac{360}{10} = 36$$

Daher muss man sich immer 36° drehen: `rt 36`.

Probieren wir das, indem wir das folgende Programm schreiben:

```
repeat 10 [ fd 50 rt 36 ]
```

Aufgabe 4.1 Zeichne folgende regelmäßige Vielecke:

- Ein 5-Eck mit der Seitenlänge 180
- Ein 12-Eck mit der Seitenlänge 50
- Ein 4-Eck mit der Seitenlänge 200
- Ein 6-Eck mit der Seitenlänge 100
- Ein 3-Eck mit der Seitenlänge 200
- Ein 18-Eck mit der Seitenlänge 20

Wenn man ein 7-Eck zeichnen will, hat man das Problem, dass man 360 nicht ohne Rest durch sieben teilen kann. In diesem Fall lässt man das Resultat durch den Rechner

ausrechnen, indem man $360/7$ schreibt. Das Symbol „/“ bedeutet für den Rechner „teile“. Der Rechner findet dann das hinreichend genaue Resultat. Somit kann man ein 7-Eck mit Seitenlänge 100 wie folgt zeichnen.

```
repeat 7 [ fd 100 rt 360/7 ]
```

Aufgabe 4.2 Zeichne folgende regelmäßige Vielecke:

- a) Ein 13-Eck mit der Seitenlänge 30
- b) Ein 19-Eck mit der Seitenlänge 20

Wir haben jetzt gelernt, regelmäßige Vielecke zu zeichnen, aber wie zeichnet man einen Kreis? Mit den Befehlen `fd` und `rt` kann man keine genauen Kreise zeichnen. Wie du aber sicherlich beobachtet hast, sehen Vielecke mit vielen Ecken Kreisen sehr ähnlich. Wenn wir also viele Ecken und sehr kurze Seiten nehmen, erhalten wir dadurch Kreise.

Aufgabe 4.3 Untersuche die Auswirkungen folgender Programme:

- a) `repeat 360 [fd 1 rt 1]`
- b) `repeat 180 [fd 3 rt 2]`
- c) `repeat 360 [fd 2 rt 1]`
- d) `repeat 360 [fd 3.5 rt 1]` (3.5 bedeutet dreieinhalb)

Aufgabe 4.4 Was würdest du tun, um ganz kleine Kreise zu zeichnen? Schreibe ein Programm dafür.

Aufgabe 4.5 Was würdest du tun, um große Kreise zu zeichnen? Schreibe ein Programm dafür.

Aufgabe 4.6 Versuche die Halbkreise aus Abb. 4.2 zu zeichnen.



(a)



(b)

Abbildung 4.2

Wir haben schon einiges gelernt und können mehrere geometrische Figuren zeichnen. Damit kann man schon sehr schöne Fantasiemuster zeichnen. Eine Idee dazu zeigen wir jetzt. Zeichne ein 7-Eck mit

```
repeat 7 [ fd 100 rt 360/7 ],
```

dann drehe die Schildkröte um 10 Grad mit

```
rt 10
```

und wiederhole

```
repeat 7 [ fd 100 rt 360/7 ].
```

Mache das ein paar Mal und schaue dir das Bild an. Wir drehen nach jedem 7-Eck immer um 10 Grad mit `rt 10`. Wenn wir wieder in die Ausgangsrichtung zurückkommen wollen, dann müssen wir diese Tätigkeit

$$\frac{360}{10} = 36$$

Mal wiederholen. Also schauen wir uns an, was das folgende Programm zeichnet:

```
repeat 36 [ repeat 7 [ fd 100 rt 360/7 ] rt 10 ].
```

Aufgabe 4.7 Zeichne ein regelmäßiges 12-Eck mit Seiten der Länge 70 und drehe es 18-mal bis du wieder an die Startposition kommst. Hinweis: Du kannst zuerst ein Programm für ein 12-Eck mit Seitenlänge 70 schreiben und ihm zum Beispiel den Namen `ECK12` geben. Dann musst du nur noch das Programm vervollständigen:

```
repeat 18 [ ECK12 rt ( was muss hier stehen? ) ].
```

Aufgabe 4.8 Denke dir eine ähnliche Aufgabe wie in Aufgabe 4.7 aus und schreibe ein Programm dazu.

Aufgabe 4.9 Ersetze das 12-Eck aus Aufgabe 4.7 durch einen Kreis (als 360-Eck mit Seitenlänge 2) und schaue dir das gezeichnete Muster an.

Wenn man schon Fantasiemuster zeichnet, passen dazu auch Farben. Die Schildkröte kann nicht nur mit Schwarz, sondern mit einer beliebigen Farbe zeichnen. Jede Farbe ist durch eine Zahl bezeichnet. Eine Übersicht aller Farben findest du in Tabelle 4.1.






Farbnummer	Farbname	[R G B]	Farbe
0	black	[0 0 0]	
1	red	[255 0 0]	
2	green	[0 255 0]	
3	yellow	[255 255 0]	
4	blue	[0 0 255]	
5	magenta	[255 0 255]	
6	cyan	[0 255 255]	
7	white	[255 255 255]	
8	gray	[128 128 128]	
9	lightgray	[192 192 192]	
10	darkred	[128 0 0]	
11	darkgreen	[0 128 0]	
12	darkblue	[0 0 128]	
13	orange	[255 200 0]	
14	pink	[255 175 175]	
15	purple	[128 0 255]	
16	brown	[153 102 0]	

Tabelle 4.1 Tabelle der Farben

Hinweis für die Lehrperson Die Spalten in Tab. 4.1 entsprechen den drei Möglichkeiten in XLOGO, eine gewünschte Farbe einzustellen. Alle Möglichkeiten sind gleichwertig. Es spielt keine Rolle, welche Farbenbezeichnung man wählt. Eine allgemeine internationale Bezeichnung, das RGB-Modell, ist in der Spalte 3 dargestellt und gibt an, wie man die gewünschte Farbe aus einer Mischung der Farben Rot, Grün und Blau erhalten kann. Auf der WEB-Seite <http://de.wikipedia.org/wiki/RGB-Farbraum> findet man eine ausführliche Erklärung. Ein Vorteil der Bezeichnung von Farben im [r,g,b]-Format ist die Tatsache, dass man hiermit beliebige Farben mischen kann, also auch solche, die in Tabelle 4.1 nicht vorkommen.

In SUPERLOGO gibt es hingegen nur zwei Möglichkeiten, Farben zu beschreiben: entweder durch eine Zahl (erste Spalte) oder durch die RGB-Zahlentripel (dritte Spalte). Dabei ist die Nummerierung der Farben in der ersten Spalte anders als bei XLOGO.

Mit dem Befehl

<u>setpencolor</u>	<u>X</u>
setzte die Farbe	Eine Zahl als Parameter zur Farbbestimmung

wechselt die Schildkröte von der aktuellen Farbe in die Farbe mit dem Wert *X*. Statt eine Zahl zu schreiben, kannst du auch direkt die gewünschte Farbe durch ihren Namen (Tabelle 4.1) angeben. Somit darfst du zum Beispiel den Befehl `setpencolor darkblue` verwenden.

Damit kann man tolle Muster zeichnen, wie zum Beispiel das Muster, das durch das folgende Programm entsteht. Zuerst benennen wir zwei Programme zum Zeichnen zweier Kreise mit unterschiedlicher Größe.

```
to KREIS3
repeat 360 [ fd 3 rt 1 ]
end

to KREIS1
repeat 360 [ fd 1 rt 1 ]
end
```

Jetzt nutzen wir diese Kreise, um ähnliche Muster wie die Bisherigen zu entwerfen.

```
to MUST3
repeat 36 [ KREIS3 rt 10 ]
end

to MUST1
repeat 18 [ KREIS1 rt 20 ]
end
```

Jetzt versuchen wir es mit Farben.

```
setpencolor 2
MUST3 rt 2
setpencolor 3
MUST3 rt 2
```

```

setpencolor 4
MUST3 rt 2
setpencolor 5
MUST3 rt 2

setpencolor 6
MUST1 rt 2
setpencolor 15
MUST1 rt 2

setpencolor 8
MUST1 rt 2
setpencolor 9
MUST1 rt 2

```

Du darfst gerne die Arbeit fortsetzen und noch mehr dazu zeichnen. Oder zeichne ein Muster nach eigener Vorstellung.

Aufgabe 4.10 Nutze `MUST3`, um das entsprechende Bild mit Orange zu zeichnen. Verwende danach den Befehl `setpencolor 7`, um zur weißen Farbe zu wechseln. Was passiert jetzt, wenn du wieder `MUST3` ausführen lässt?

Aufgabe 4.11 Zeichne das Bild in Abb. 4.3. Die Schildkröte ist am Anfang an dem gemeinsamen Punkt (in dem Schnittpunkt) der beiden Kreise.




Abbildung 4.3

Zusammenfassung

Wir haben gelernt, wie man regelmäßige Vielecke zeichnen kann. Man muss Strecken mit gleicher Länge zeichnen und zwischen dem Zeichnen der Seiten immer mit

```
rt 360/Anzahl der Ecken
```

drehen. Einen Kreis zeichnet man als ein Vieleck mit sehr vielen Ecken, am besten mit 360 oder 180.

Die Schildkröte kann mit beliebigen Farben zeichnen. Um die Farbe zu wechseln, verwendet man das Befehlswort `setpencolor` und als Parameter kommt danach die Nummer der gewünschten Farbe.

Kontrollfragen

1. Wie zeichnet man regelmäßige Vielecke? Was hat die Anzahl der Ecken mit der Größe der Drehung nach dem Zeichnen einer Seite zu tun?
2. Wie berechnet man den Umfang eines Vielecks?
3. Wie zeichnet man Kreise?
4. Mit welchem Befehl kann man die Stiftfarbe der Schildkröte ändern?
5. Kann man bei der Schildkröte durch Farbänderung den Stiftmodus ändern? Wenn ja, wie und in welchen Modus?

Kontrollaufgaben

1. Zeichne folgende regelmäßige Vielecke:
 - a) Ein 12-Eck mit Seitenlänge 25
 - b) Ein 7-Eck mit Seitenlänge 50
 - c) Ein 3-Eck mit Seitenlänge 200

Bestimme für alle den Umfang.

2. Zeichne Kreise mit folgenden Umfängen:
 - a) 360 Schritte
 - b) 720 Schritte

c) 900 Schritte

d) 777 Schritte

3. Schreibe ein Programm zum Zeichnen des Bildes aus Abb. 4.4. Der Umfang der Kreise ist jeweils 540 Schritte.




Abbildung 4.4

4. Schreibe ein Programm zum Zeichnen des Bildes aus Abb. 4.5. Die Größe der Kreise darfst du selbst wählen.




Abbildung 4.5

5. Schreibe ein Programm zum Zeichnen des Bildes aus Abb. 4.6.




Abbildung 4.6

6. Kannst du mit Hilfe des Radiergummimodus und des Stiftmodus das Bild aus Abb. 4.6 auf der vorherigen Seite in vier nebeneinander stehende Häuser umwandeln? Wie die Häuser aussehen sollen, ist dir überlassen. Ein Beispiel für ein Haus siehst du in Abb. 4.7.




Abbildung 4.7

7. Zeichne mit Gelb das Netz aus Abb. 4.8.




Abbildung 4.8

8. Zeichne vier Kreise wie in Abb. 4.9 mit den Umfängen 360, 540, 720 und 900.




Abbildung 4.9

Lösungen zu ausgesuchten Aufgaben

Aufgabe 4.2

a) `repeat 13 [fd 30 rt 360/13]`

b) `repeat 19 [fd 20 rt 360/19]`

Aufgabe 4.5

Ein Kreis wird als ein Vieleck mit vielen Ecken gezeichnet. Damit ist sein Umfang immer *Anzahl der Ecken* \times *Seitenlänge*. Wenn wir Kreise als regelmäßige 360-Ecke zeichnen, ist der Umfang $360 \times$ *Seitenlänge*. Somit ist der Umfang des Kreises

```
repeat 360 [ fd 1 rt 1 ]
```

$360 \times 1 = 360$ Schritte. Wenn wir den Umfang vergrößern, dann können wir die Seitenlänge vergrößern. Zum Beispiel zeichnet

```
repeat 360 [ fd 4 rt 1 ]
```

einen Kreis mit dem Umfang $360 \times 4 = 1440$.

Aufgabe 4.6

Einen Halbkreis zeichnet man, indem man statt 360-mal nur 180-mal mittels `rt 1` dreht. Wenn die Schildkröte für das Bild aus Abb. 4.2(a) am Anfang auf der linken Seite steht, reicht das folgende Programm:

```
repeat 180 [ fd 2 rt 1 ]
```

Nach dem Zeichnen steht die Schildkröte am rechten Ende des Halbkreises und schaut nach unten.

Um den Halbkreis in Abb. 4.2(b) zu zeichnen, können wir unterschiedlich vorgehen. Zum Beispiel können wir den Halbkreis in Abb. 4.2(b) als den zweiten Teil eines Kreises sehen, bei dem der Halbkreis in Abb. 4.2(a) dem ersten Teil entspricht. Wir können also den ersten Teil ablaufen, ohne ihn zu zeichnen und danach den zweiten Teil zeichnen. Das macht das folgende Programm:

```
pu
repeat 180 [ fd 2 rt 1 ]
pd
repeat 180 [ fd 2 rt 1 ]
```

Wir können auch anders vorgehen. Wir drehen die Schildkröte um und lassen sie den Halbkreis aus Abb. 4.2(b) als den ersten Teil eines Kreises zeichnen.

```
rt 180
repeat 180 [ fd 2 rt 1 ]
```

Kontrollaufgabe 2

- b) `repeat 360 [fd 2 rt 1]`
- c) `repeat 360 [fd 2.5 rt 1]`
- d) `repeat 360 [fd 777/360 rt 1]`

Kontrollaufgabe 5

Wir zeichnen das Bild aus Abb. 4.6 auf Seite 81 von links nach rechts. Wir gehen dabei modular vor und schreiben zuerst zwei Programme. Das erste Programm zeichnet ein 6-Eck mit der Seitenlänge 100.

```
to ECK6L100
repeat 6 [ fd 100 rt 60 ]
end
```

Das zweite Programm macht die notwendige Verschiebung zur neuen Startposition, von wo aus das nächste 6-Eck gezeichnet werden kann.

```
to VERS
repeat 4 [ fd 100 rt 60 ]
rt 120
end
```

Damit sieht das Programm zum Zeichnen des Bildes aus Abb. 4.6 auf Seite 81 wie folgt aus:

```
repeat 3 [ ECK6L100 VERS ]
ECK6L100.
```